

# Program WinSpice

Jan Bičák

30. října 2007

## Abstrakt

Tento text představuje velmi stručný manuál k programu WinSpice. Je vytvořen pro podporu výuky předmětu Elektronické obvody pro sdělovací techniku - EOS, vyučovaného na katedře teorie obvodů, FEL ČVUT.

## 1 Úvod

WinSpice3 je simulační program pro analýzu elektrických a elektronických obvodů. Umožňuje provádět následující typy analýz:

- nelineární stejnosměrná (DC),
- nelineární přechodová (TRAN) a
- lineární střídavá (AC).

Obvody mohou obsahovat rezistory, kapacitory, induktory, vázané induktory, nezávislé zdroje napětí a proudu, čtyři typy řízených zdrojů, bezztrátová a ztrátová přenosová vedení, spínače a pět typů nejvíce používaných polovodičových prvků: diody, bipolární tranzistory, unipolární tranzistory MOSFET, JFET a MESFET (GaAs FET).

WinSpice3 je založen na programu Spice3F4<sup>1</sup> <sup>2</sup>, který byl zase vyvinut z programu SPICE2G.6. Proti výše uvedeným programům má WinSpice3 implementovány některé nové funkce (více [1]). Program WinSpice3 je shareware a je možné si ho stáhnout zdarma z internetu (viz. [2]). Současná sharewareová verze programu není nijak omezena ve funkčnosti (jako např. maximální počet uzlů, součástek ap.). Samotný program WinSpice3 nemá schematický editor a popis obvodu je zadáván jako textový soubor, jak bude ukázáno v další části.

## 2 Popis obvodu - vstupní soubor

Vlastní obvod, který má být analyzován, musí být zadán ve formě textové souboru s příponou ".cir". První řádek v souboru je brán jako popis obvodu (souboru) a poslední řádek musí být ".END". Každá součástka obvodu je pak zadána na samostatné řádce, která začíná jménem součástky, dále obsahuje uzly obvodu, ke kterým je součástka připojena a nakonec jsou uvedeny hodnoty parametrů, které specifikují elektrické vlastnosti součástky. První písmeno jména součástky určuje typ součástky (např. jména R1, Rout značí rezistory, V1 značí nezávislý zdroj napětí atd.). Každá součástka obvodu musí mít unikátní jméno. Uzly obvodu se mohou značit jak číslem tak i alfanumerickým řetězcem. Referenční uzel se značí jako číslo 0. Pro označení numerických hodnot lze využít značení, které je uvedeno v tab. 1. Místo velkých písmen lze psát i malá písmena. Pokud se deklarace nevejde na jeden řádek, je možné pokračovat na dalších řádcích. Každý pokračovací řádek musí začínat znakem plus "+". V souboru je možné použít prázdné řádky. Řádek, který začíná znakem hvězdičky "\*", se bere jako komentář. V celém souboru je nutné se vyvarovat použití diakritiky. WinSpice při nalezení znaku s diakritikou ohlásí chybu a to i na řádku komentáře. Toto

<sup>1</sup>Spice3F4 byl vyvinut na katedře Electrical Engineering and Computer Sciences, University of California, Berkeley.

<sup>2</sup>Spice je akronym pro Simulation Program with Integrated Circuit Emphasis.

platí pro verzi 1.5.7, které je dostupná v době psaní tohoto textu. Příklad vstupního souboru je uveden na obr. 1. Soubor obsahuje také řídicí část s příkazy pro analýzu a zobrazení výsledků uvedenou klíčovým slovem ".CONTROL" a zakončenou ".ENDC". Tyto příkazy budou popsány v další části. Pokud soubor otevřeme v programu WinSpice, tak program provede analýzu (výpočet stejnosměrného pracovního bodu) a vypíše výsledek (všechna napětí v uzlech a proud zdrojem napětí). Výstup programu je uveden na obr. 2.

násobnost	symbol	název
$10^{-15}$	F	femto
$10^{-12}$	P	pico
$10^{-9}$	N	nano
$10^{-6}$	U	micro
$10^{-3}$	M	mili
$10^3$	K	kilo
$10^6$	MEG	mega
$10^9$	G	giga
$10^{12}$	T	tera

Tabulka 1: Značení řádů

---

```

Delic napeti - prvni radek

* Popis obvodu - komentar
V1 1 0 10
R1 1 2 1K
R2 2 0 2K

* Ridici cast
.CONTROL
OP
PRINT ALL
.ENDC

.END

```

---

Obrázek 1: Příklad vstupního souboru pro WinSpice

## 2.1 Obvodové prvky a modely

Tab. 2 ukazuje přehled všech základních prvků, které je možné použít v programu WinSpice3. V další části bude uvedena stručná syntaxe zadávání obvodových prvků a modelů. Jsou uvedeny pouze ty obvodové prvky a jejich parametry, které považují za důležité z hlediska předmětu EOS. Parametry uvedené mezi znaky < a > jsou volitelné.

### 2.1.1 Rezistor

Obecný tvar:

```

RXXXXXXXX N1 N2 VALUE
RXXXXXXXX N1 N2 R=<expresion>

```

Příklady:

Circuit: Delic napeti - prvni radek

TEMP=27 deg C  
DC Operating Point ... 100%

v(1) = 1.000000e+01  
v(2) = 6.666667e+00  
v1#branch = -3.33333e-03

Obrázek 2: Výsledek po zpracování vstupního souboru

písmeno	popis
R	rezistor
C	kapacitor
L	induktor
K	vázaný induktor
S	spínač řízený napětím
W	spínač řízený proudem
I	nezávislý zdroj proudu
G	lineární zdroj proudu řízený napětím
E	lineární nebo nelineární zdroj napětí řízený napětím
F	lineární nebo nelineární zdroj proudu řízený proudem
H	lineární zdroj napětí řízený proudem
B	nelineární řízený zdroj napětí nebo proudu
T	bezeztrátové přenosové vedení
O	ztrátové přenosové vedení
U	RC přenosové vedení
D	dioda
Q	bipolární tranzistor
J	JFET tranzistor
M	MOSFET tranzistor
Z	MESFET tranzistor

Tabulka 2: Základní prvky

```
R1 1 2 100  
RC1 12 17 1K  
RC2 4 5 R=1000+log(v(1))  
Rbot 8 0 R=1000+1000*sin(2*3.14159*10000*time)
```

N1 a N2 značí dva uzly prvku. VALUE je odpor v ohmech a může být kladný nebo záporný, ale nesmí být nulový. Hodnota odporu může být zadána také jako výraz (**expression**), jak ukazuje třetí a čtvrtý příklad. Syntaxe zadávání teplotní závislosti rezistorů a polovodičových rezistorů je uvedena v [1].

### 2.1.2 Kapacitor

Obecný tvar:

```
CXXXXXXXX N+ N- VALUE <IC=VAL>  
CXXXXXXXX N+ N- C=<expression>
```

Příklady:

```
CBYP 13 0 1U
COSC 17 23 10U IC=3
C2 5 6 C=10u
```

N+ a N- značí kladný resp. záporný uzel prvku. VALUE je kapacita ve faradech. IC představuje volitelnou počáteční podmínku jako počáteční hodnotu napětí na kapacitoru v čase 0. Tato hodnota má význam pro přechodovou analýzu TRAN, při použití parametru UIC. Hodnota kapacity může být zadána také jako výraz (*expression*). Syntaxe zadávání polovodičových kapacitorů je uvedena v [1].

### 2.1.3 Induktor

Obecný tvar:

```
LYYYYYYY N+ N- VALUE <IC=INCOND>
LYYYYYYY N+ N- L=<expression>
```

Příklady:

```
LLINK 42 69 1U
LLINK 42 69 L=1U
LSHUNT 23 51 10u IC=15.7m
```

N+ a N- značí kladný resp. záporný uzel prvku. VALUE je indukčnost v henry. IC představuje volitelnou počáteční podmínku jako počáteční hodnotu proudu, který teče v čase 0 přes induktor od uzlu N+ k uzlu N-. Tato hodnota má význam pro přechodovou analýzu TRAN, při použití parametru UIC. Hodnota indukčnosti může být zadána také jako výraz (*expression*).

### 2.1.4 Vázaný induktor

Obecný tvar:

```
KXXXXXXX LYYYYYYY LZZZZZZZ VALUE
```

Příklady:

```
K43 LAA LBB 0.999
KXFRMR L1 L2 0.87
```

LYYYYYYY a LZZZZZZZ jsou jména dvou vázaných induktorů a VALUE je činitel vazby  $K$ , který musí být větší než 0 a menší nebo roven 1. Při značení orientace pomocí teček, je tečka umístěna na kladném uzlu každého induktoru. Mezi činitelem vazby  $K$  a vzájemnou indukčností  $M$  platí vztah

$$K = \frac{M}{\sqrt{L_1 L_2}}$$


### 2.1.5 Nezávislý zdroj napětí nebo proudu

Obecný tvar:

```
VXXXXXXX N+ N- <<DC> DC_VALUE> <AC <AC_MAG <AC_PHASE>>> <TRAN_FUNC>
IXXXXXXXX N+ N- <<DC> DC_VALUE> <AC <AC_MAG <AC_PHASE>>> <TRAN_FUNC>
```

Příklady:

```
VCC 10 0 DC 6
VIN 13 2 0 AC 1 SIN(0 1 1MEG)
ISRC 23 21 DC 0.333 AC 0.333 45.0
VMEAS 12 9
```



Obrázek 3: Průběh funkce PULSE

N+ a N- značí kladný resp. záporný uzel prvku. Kladný proud teče od uzlu N+ přes zdroj k uzlu N-. Napěťové zdroje s nulovým napětím lze využít jako ampérmetry. Svým chováním představují v obvodu zkrat a WinSpice při analýze vypočítá proud, který přes ně teče.

DC\_VALUE představuje stejnosměrnou hodnotu zdroje pro stejnosměrnou analýzu. Pokud není uvedena TRAN\_FUNC tak je tato hodnota použita i pro přechodovou analýzu. Pokud není hodnota DC\_VALUE uvedena, tak se jako stejnosměrná hodnota použije hodnota v čase 0 podle TRAN\_FUNC. Pokud není uvedena ani TRAN\_FUNC, tak se předpokládá stejnosměrná hodnota rovna 0.

AC\_MAG a AC\_PHASE představují amplitudu a fázi ve stupních pro střídavou analýzu. Pokud je uveden pouze identifikátor AC, tak se použije amplituda 1 a fáze 0.

TRAN\_FUNC představuje časově proměnou funkci pro přechodovou analýzu. Je možné použít pět různých funkcí: sinusovou SIN, pulsní PULSE, exponenciální EXP, po částech lineární PWL a frekvenčně modulovaný SFFM. V další popisu jsou použity dvě konstanty podle přechodové analýzy TSTEP a TSTOP. TSTEP představuje časový interval použitý pro tisknutí výsledků z přechodové analýzy a TSTOP představuje čas, do kterého se přechodová analýza počítá.

### Pulsní funkce PULSE

Obecný tvar:

PULSE(V1 V2 <TD <TR <TF <PW <PER>>>>>>)

Příklady:

V1 1 0 PULSE(0 1)

VIN 3 0 PULSE(-1 1 2N 2N 2N 50N 100N)

V tab. 3 je uveden význam jednotlivých parametrů a na obr. 3 je průběh funkce.

Označení	Význam	Jednotky	Default
V1	Počáteční hodnota	V nebo A	0.0
V2	Pulsní hodnota	V nebo A	0.0
TD	Počáteční prodleva	s	0.0
TR	Doba náběžné hrany	s	TSTEP
TF	Doba sestupné hrany	s	TSTEP
PW	Šířka pulsu	s	TSTOP
PER	Perioda	s	TSTOP

Tabulka 3: Parametry funkce PULSE

## Sinusová funkce SIN

Obecný tvar:

SIN(VO VA <FREQ <TD <THETA>>>)

Příklady:

V1 1 0 SIN(0 1)

VIN 3 0 SIN(0 1 1k)

I2 2 0 SIN(0 1 10k 0 10k)

V tab. 4 je uveden význam jednotlivých parametrů. Zdroj má hodnotu stejnosměrné složky v čase

Označení	Význam	Jednotky	Default
VO	Stejnosemřná složka	V nebo A	0.0
VA	Amplituda	V nebo A	0.0
FREQ	Frekvence	Hz	1/TSTOP
TD	Zpoždění	s	0.0
THETA	Činitel útlumu	s <sup>-1</sup>	0.0

Tabulka 4: Parametry funkce SIN

od  $t = 0$  do  $t = TD$ . Pro čas  $t > TD$  se chová podle funkce

$$f(t) = VO + VA \sin(2\pi FREQ(t - TD))e^{-THETA(t-TD)}$$

## Exponenciální funkce EXP

Obecný tvar:

EXP(V1 V2 <TD1 TAU1 TD2 <TAU2>>)

Příklady:

V1 1 0 EXP(0 1)

Vin 3 0 EXP (0 1 0 100u 1m)

I2 2 0 0 EXP (1 0 0 100u 1m)

V tab. 5 je uveden význam jednotlivých parametrů. Zdroj se chová podle funkce

Označení	Význam	Jednotky	Default
V1	Počáteční hodnota	V nebo A	0.0
V2	Velikost pulsu	V nebo A	0.0
TD1	Prodleva náběžné hrany	s	0.0
TAU1	Časová konstanta náběžné hrany	s	TSTEP
TD2	Prodleva sestupné hrany	s	TD1 + TSTEP
TAU2	Časová konstanta sestupné hrany	s	TSTEP

Tabulka 5: Parametry funkce EXP

$$f(t) = \begin{cases} v_1(t) = V1 & \text{pro } 0 \leq t \leq TD1 \\ v_2(t) = v_1(t) + (V2 - V1)(1 - e^{-(t-TD1)/TAU1}) & \text{pro } TD1 < t \leq TD2 \\ v_3(t) = v_2(t) + (V1 - V2)(1 - e^{-(t-TD2)/TAU2}) & \text{pro } TD2 < t \leq TSTOP \end{cases}$$

## Po částech lineární funkce PWL

Obecný tvar:

PWL(T1 V1 <T2 V2 <T3 V3 <T4 V4 ...>>>)

Příklady:

VCLOCK 7 5 PWL(0 -7 10NS -7 11NS -3 17NS -3 18NS -7 50NS -7)

Každá dvojice hodnot  $T_i$  a  $V_i$  udává, že napětí nebo proud v čase  $T_i$  je  $V_i$ . Hodnoty v mezilehlém čase jsou získány lineární interpolací.

## Frekvenčně modulovaná sinusová funkce SFFM

Obecný tvar:

SFFM(VO VA <FC <MDI <FS>>>)

Příklady:

V1 12 0 SFFM(0 1M 20K 5 1K)

V tab. 6 je uveden význam jednotlivých parametrů. Zdroj se chová podle funkce

$$f(t) = VO + VA \sin(2\pi FC t + MDI \sin(2\pi FS t)).$$

Označení	Význam	Jednotky	Default
VO	Stejnoseměrná složka	V nebo A	0.0
VA	Amplituda	V nebo A	0.0
FC	Frekvence nosné	Hz	1/TSTOP
MDI	Modulační index	-	0.0
FS	Frekvence signálu	Hz	1/TSTOP

Tabulka 6: Parametry funkce SFFM

### 2.1.6 Lineární řízené zdroje

Obvody analyzované v programu SPICE mohou obsahovat lineární řízené zdroje, které lze popsat čtyřmi rovnicemi

$$i = g u, \quad u = e i, \quad i = f u, \quad u = h i,$$

kde  $g$  představuje transkonduktanci,  $e$  napěťové zesílení,  $f$  proudové zesílení a  $h$  představuje transrezistanci.

### Zdroj proudu řízený napětím

Obecný tvar:

GXXXXXXX N+ N- NC+ NC- VALUE

Příklad:

G1 2 0 5 0 0.1M

N+ a N- značí kladný resp. záporný uzel prvku. Kladný proud teče od uzlu N+ přes zdroj k uzlu N-. NC+ a NC- značí kladný a záporný řídicí uzel. VALUE je transkonduktance.

## Zdroj napětí řízený napětím

Obecný tvar:

```
EXXXXXXX N+ N- NC+ NC- GAIN
```

Příklad:

```
E1 2 3 14 1 2.0
```

N+ a N- značí kladný resp. záporný uzel prvku. NC+ a NC- značí kladný a záporný řídicí uzel. GAIN je napěťové zesílení.

## Zdroj proudu řízený proudem

Obecný tvar:

```
FXXXXXXX N+ N- VNAM GAIN
```

Příklad:

```
F1 13 5 VSENS 5
```

N+ a N- značí kladný resp. záporný uzel prvku. Kladný proud teče od uzlu N+ přes zdroj k uzlu N-. VNAM je jméno napěťového zdroje, přes který teče řídicí proud. Kladný řídicí proud teče od kladného uzlu přes zdroj k zápornému uzlu zdroje VNAM. GAIN je proudové zesílení.

## Zdroj napětí řízený proudem

Obecný tvar:

```
HXXXXXXX N+ N- VNAM VALUE
```

Příklad:

```
HX 5 17 VZ 0.5K
```

N+ a N- značí kladný resp. záporný uzel prvku. VNAM je jméno napěťového zdroje, přes který teče řídicí proud. Kladný řídicí proud teče od kladného uzlu přes zdroj k zápornému uzlu zdroje VNAM. VALUE je transrezistance.

### 2.1.7 Dioda

Obecný tvar:

```
DXXXXXXX N+ N- MNAME <AREA> <OFF> <IC=VD>
```

Příklady:

```
DBRIDGE 2 10 DIODE1
```

```
DCLMP 3 7 DMOD 3.0 IC=0.2
```

N+ značí kladný uzel (anodu) a N- značí záporný uzel (katodu). MNAME značí jméno modelu. AREA faktor je roven počtu stejných diod zapojených paralelně. Defaultová hodnota je 1.0. Parametr OFF nastavuje počáteční stav diody pro stejnosměrnou (DC) analýzu. Pokud není uvedeno, tak se předpokládá, že dioda vede a je na ní napětí  $v_D \approx 0.6$ . IC=VD představuje volitelnou počáteční podmínku jako počáteční hodnotu napětí na diodě v čase 0. Tato hodnota má význam pro přechodovou analýzu TRAN, při použití parametru UIC.



## Model diody (D)

Obecný tvar:

```
.MODEL DIODEname D(<<is=10f> <rs=1> ...)>
```

Příklady:

```
.MODEL DIODE1 D
.MODEL 1N3893 D(is=15.03f rs=9.606m n=1 tt=369.9n
+cjo=117.8p vj=.75 m=.2976 eg=1.11 xti=3 fc=.5)
```

Tab. 7 uvádí přehled nejdůležitějších parametrů modelu diody. Kompletní přehled je uveden např. v [1].

Označení	Význam	Jednotky	Default
IS	Saturační proud	A	$10^{-14}$
N	Emisní koeficient	-	1
RS	Sériový odpor	$\Omega$	0
TT	Doba průletu	s	0
CJO	Bariérová kapacita při $v_D = 0$	F	0
BV	Průrazné napětí v závěrném směru	V	$\infty$ (BV=0)
IBV	Proud při průrazném napětí	A	$10^{-3}$

Tabulka 7: Parametry modelu diody D

## 2.1.8 Bipolární tranzistor

Obecný tvar:

```
QXXXXXXX NC NB NE <NS> MNAME <AREA> <OFF> <IC=VBE, VCE>
```

Příklady:

```
Q23 10 24 13 QMOD IC=0.6, 5.0
Q50A 11 26 4 20 MOD1
```

NC, NB a NE značí postupně uzly kolektoru, báze a emitoru. NS značí uzel substrátu (není nutné zadávat). MNAME značí jméno modelu. AREA faktor je roven počtu stejných tranzistorů zapojených paralelně. Defaultová hodnota je 1.0. Parametr OFF nastavuje počáteční stav tranzistoru pro stejnosměrnou (DC) analýzu. Pokud není uvedeno, tak se předpokládá, že je tranzistor v aktivní oblasti s napětím  $U_{BE} = 0.6V$  a s napětím  $U_{BC} = -1.0V$ . Parametr IC=VBE,VCE představuje volitelnou počáteční podmínku jako počáteční hodnoty napětí v čase 0. Tato hodnota má význam pro přechodovou analýzu TRAN, při použití parametru UIC.

## Model bipolárního tranzistoru (NPN, PNP)

Obecný tvar:

```
.MODEL TRANSISTORname NPN(<<is=1.0e-16> <bf=100> ...)>
.MODEL TRANSISTORname PNP(<<is=1.0e-16> <bf=100> ...)>
```

Příklady:

```
.MODEL QMOD NPN
.MODEL BC337-16 NPN(is=3.941445e-14 bf=160 nf=1 vaf=109.45 ikf=.8 ise=7.4025e-15
+ne=1.3 br=20.5 nr=.974 var=14.25 ikr=.1 isc=3.157e-13 nc=1.2 rb=1.1 re=.1259
+rc=.0539 cje=63e-12 tf=.75e-9 cjc=15.8e-12 vjc=.505 mjc=.39 tr=85e-9 )
```

Tab. 8 uvádí přehled nejdůležitějších parametrů modelu bipolárního tranzistoru. Kompletní přehled je uveden např. v [1].

Označení	Význam	Jednotky	Default
IS	Saturační proud	A	$10^{-16}$
BF	Proudové zesílení v aktivním režimu	-	100
BR	Proudové zesílení v inverzním režimu	-	1
NF	Emisní koeficient přechodu B-E	-	1
NR	Emisní koeficient přechodu B-C	-	1
VAF	Earlyho napětí v aktivním režimu	V	$\infty$ (VAF=0)
VAR	Earlyho napětí v inverzním režimu	V	$\infty$ (VAR=0)
RC	Sériový odpor kolektoru	$\Omega$	0
RE	Sériový odpor emitoru	$\Omega$	0
RB	Sériový odpor báze	$\Omega$	0
TF	Doba průletu přechodu B-E	s	0
TR	Doba průletu přechodu B-C	s	0
CJE	Bariérová kapacita přechodu B-E při $v_{BE} = 0$	F	0
CJC	Bariérová kapacita přechodu B-C při $v_{BC} = 0$	F	0

Tabulka 8: Parametry modelu tranzistoru Q

### 2.1.9 Tranzistor MOSFET

Obecný tvar:

```
MXXXXXXX ND NG NS NB MNAME <L=VAL> <W=VAL> <AD=VAL> <AS=VAL>
+ <PD=VAL> <PS=VAL> <NRD=VAL> <NRS=VAL> <OFF>
+ <IC=VDS, VGS, VBS> <TEMP=T>
```

Příklady:

```
M1 24 2 0 20 TYPE1
M31 2 17 6 10 MODM L=5U W=2U
```

ND, NG, NS a NB značí postupně uzly drain, gate, source a bulk (substrát). MNAME značí jméno modelu. Parametry L a W značí délku a šířku kanálu v metrech. Parametr OFF nastavuje počáteční stav tranzistoru pro stejnosměrnou (DC) analýzu. Pokud není uvedeno, tak se předpokládá, že je tranzistor na hranici otevření s napětím  $U_{GS} = U_{TO}$ ,  $U_{DS} = 0.0V$  a  $U_{BS} = -1.0V$ . Parametr IC=VDS, VGS, VBS představuje volitelnou počáteční podmínku jako počáteční hodnoty napětí v čase 0. Tato hodnota má význam pro přechodovou analýzu TRAN, při použití parametru UIC.

### Model tranzistoru MOSFET (NMOS, PMOS)

Obecný tvar:

```
.MODEL TRANSISTORname NMOS(<vto=0.0> <kp=2e-5> ...)
.MODEL TRANSISTORname PMOS(<vto=0.0> <kp=2e-5> ...)
```

Příklady:

```
.MODEL TYPE1 NMOS(level=1 vto=1.5 kp=3e-5)
```

U tranzistoru MOSFET existuje celá řada modelů (parametr LEVEL), které se liší úrovní a složitostí modelování chování tranzistoru MOSFET. Nejjednodušší je úroveň LEVEL=1. Tab. 9 uvádí přehled nejdůležitějších parametrů modelu tranzistoru MOSFET pro LEVEL 1, 2, 3 a 6. Kompletní přehled je uveden např. v [1].

Označení	Význam	Jednotky	Default
LEVEL	Index modelu	-	1
VTO	Prahové napětí	V	0
KP	Parametr transkonduktance	$AV^{-2}$	$2 \times 10^{-5}$
LAMBDA	Parametr modulace délky kanálu	1/V	0.0

Tabulka 9: Parametry modelu tranzistoru M

### 2.1.10 Spínač řízený napětím

Obecný tvar:

SXXXXXXX N+ N- NC+ NC- MODEL <ON><OFF>

Příklady:

```
s1 1 2 3 4 switch1 ON
s2 5 6 3 0 sm2 off
Switch1 1 2 10 0 smodel1
```

N+ a N- značí kladný resp. záporný uzel prvku. Kladný proud teče od uzlu N+ přes zdroj k uzlu N-. NC+ a NC- značí kladný a záporný řídicí uzel. MODEL značí jméno modelu. Parametr ON/OFF nastavuje počáteční stav spínače pro stejnosměrnou (DC) analýzu.

### Model spínače řízeného napětím (SW)

Obecný tvar:

.MODEL SPNAME SW(PNAME1=PVAL1 PNAME2=PVAL2 ... )

Příklady:

.MODEL SMOD SW(ROFF=5M RON=10E9 VT=1.0 VH=0.1)

Tab. 10 uvádí přehled parametrů modelu spínače řízeného napětím.

Označení	Význam	Jednotky	Default
VT	Prahové napětí	V	0
VH	Hysteréze	V	0
RON	Odpor v sepnutém stavu	$\Omega$	1
ROFF	Odpor v rozepnutém stavu	$\Omega$	$1/GMIN=1T\Omega$

Tabulka 10: Parametry modelu spínače řízeného napětím SW

## 3 Analýza obvodu - Řídící příkazy - Výstup programu

### 3.1 Typy analýz

Následující přehled není kompletní. Všechny typy analýz, které je možné provádět v programu WinSpice, jsou uvedeny v [1].

#### 3.1.1 Nelineární stejnosměrná analýza

Při stejnosměrné analýze jsou induktoři nahrazeny zkratem a kapacitory rozpojeným obvodem.

## Výpočet pracovního bodu OP

Obecný tvar:

OP

Příkaz nemá žádné parametry.

## Stejnoseměrná analýza DC

Obecný tvar:

```
DC SRCNAM VSTART VSTOP VINCR <SRC2 START2 STOP2 INCR2>
```

Příklady:

```
DC Vin 0 5 0.1
```

```
DC Vce 0 10 0.2 IB 0 10u 1u
```

Provede opakovaný výpočet pracovního bodu, pokud se napětí nebo proud uvedeného zdroje SRCNAM mění v zadaném rozsahu od VSTART do VSTOP s krokem VINCR. Pokud jsou uvedeny také argumenty SRC2 START2 STOP2 INCR2, tak se tento opakovaný výpočet provádí pro každou hodnotu z rozsahu START2 STOP2, kde SRC2 určuje druhý zdroj, jehož hodnota se má měnit s krokem INCR2.

U prvního příkladu se provádí opakovaný výpočet pracovního bodu pro měnící se napětí zdroje Vin v intervalu 0 až 5V s krokem 0.1V. U druhého příkladu se provádí opakovaný výpočet pracovního bodu pro měnící se napětí zdroje Vce v intervalu 0 až 10V s krokem 0.2V a to pro každou hodnotu zdroje proudu IB, která se také mění a to v intervalu 0 až 10 $\mu$ A s krokem 1 $\mu$ A.

## Stejnoseměrná přenosová funkce v pracovním bodě TF

Obecný tvar:

```
TF OUTVAR INSRC
```

Příklady:

```
TF V(5, 3) VIN
```

```
TF I(VLOAD) VIN
```

Výpočet stejnosměrného přenosu pro zadanou výstupní veličinu OUTVAR a pro zadaný vstupní zdroj INSRC. Dále se provede výpočet vstupního a výstupního odporu. U nelineárního obvodu je výpočet platný pro malé změny signálu v okolí pracovního bodu.

### 3.1.2 Nelineární přechodová analýza

#### Přechodová analýza TRAN

Obecný tvar:

```
TRAN TSTEP TSTOP <TSTART <TMAX>><UIC>
```

Příklady:

```
TRAN 1N 100N
```

```
TRAN 1N 1000N 500N
```

```
TRAN 10N 1U UIC
```

Program provede přechodovou analýzu v časovém intervalu 0 až TSTOP. Pokud je uveden argument TSTART, tak jsou výsledky k dispozici v časovém intervalu od TSTART do TSTOP. Pokud není TSTART specifikován, předpokládá se, že je 0. TSTEP je požadovaný časový krok přechodové analýzy. WinSpice může podle potřeby tento krok zjemnit, aby byla zajištěna odpovídající přesnost výpočtu (podle nastavení vnitřních proměnných). Pokud je

$$TSTEP > \frac{TSTOP - TSTART}{50.0},$$

tak se jako TSTEP použije hodnota z pravé strany nerovnosti. Pokud je specifikován argument TMAX, tak je časový krok nastaven podle tohoto argumentu, bez ohledu na TSTEP a to i v případě, že není splněna výše uvedená nerovnost.

Před provedením výpočtu přechodové analýzy program provede výpočet stejnosměrného pracovního bodu OP a vypočítané hodnoty se použijí jako počáteční podmínky. Pokud je uveden argument UIC, tak se výpočet pracovního bodu neprovádí a jako počáteční podmínky se použijí hodnoty IC, které je možné specifikovat pro jednotlivé součástky, případně uzly. Výchozí hodnota pro parametry IC je 0.

### Fourierova analýza FOUR

Obecný tvar:

```
FOUR FREQ OV1 <OV2 OV3 ...>
```

Příklady:

```
FOUR 100K V(5)
```

Program provede fourierovu analýzu na základě výsledků přechodové analýzy a zobrazí výsledek ve formě tabulky. Argument FREQ udává základní frekvenci a OV1 je vektor s časovým průběhem výstupní veličiny z přechodové analýzy. Argumenty OV2 OV3 ... udávají případné další výstupní veličiny. Fourierova analýza je prováděna na intervalu  $\langle TSTOP - 1/FREQ, TSTOP \rangle$  a určí stejnosměrnou složku a prvních 9 harmonických. Dále je vypočítáno celkové harmonické zkreslení (THD). Pro maximální přesnost je třeba nastavit krok přechodové analýzy na  $1/FREQ/100$  případně méně.

### 3.1.3 Lineární střídavá analýza

U střídavé analýzy je vždy prováděn automaticky výpočet pracovního bodu a případná linearizace nelineárního obvodu. Výsledky jsou potom platné pouze pro dostatečně malé změny signálu.

### Střídavá analýza AC

Obecný tvar:

```
AC DEC ND FSTART FSTOP
AC OCT NO FSTART FSTOP
AC LIN NP FSTART FSTOP
```

Příklady:

```
AC DEC 10 1 10K
AC DEC 10 1K 100MEG
AC LIN 100 1 100HZ
```

Program provede střídavou analýzu v intervalu od FSTART do FSTOP. Parametr LIN nastaví lineární dělení stupnice a NP určuje celkový počet bodů. Parametr DEC resp. OCT nastaví logaritmické dělení stupnice a ND resp. NO určuje počet bodů na dekádu resp. na oktávu. Aby střídavá analýza měla smysl, musí mít alespoň jeden nezávislý zdroj nastavený argument AC.

## Výpočet nul a pólů PZ

Obecný tvar:

```
PZ NODE1 NODE2 NODE3 NODE4 CUR POL
PZ NODE1 NODE2 NODE3 NODE4 CUR ZER
PZ NODE1 NODE2 NODE3 NODE4 CUR PZ
PZ NODE1 NODE2 NODE3 NODE4 VOL POL
PZ NODE1 NODE2 NODE3 NODE4 VOL ZER
PZ NODE1 NODE2 NODE3 NODE4 VOL PZ
```

Příklady:

```
PZ 1 0 3 0 CUR POL
PZ 2 3 5 0 VOL ZER
PZ 4 1 4 1 CUR PZ
```

Program vypočítá nuly a póly přenosové funkce. Argumenty NODE1 a NODE2 určují vstupní uzly a argumenty NODE3 a NODE4 určují výstupní uzly. Pro argument CUR je přenosová funkce typu

$$\frac{\text{výstupní napětí}}{\text{vstupní proud}}$$

a pro VOL je typu

$$\frac{\text{výstupní napětí}}{\text{vstupní napětí}}$$

Argument POL resp. ZER určuje, že budou vypočítány pouze póly resp. nuly. Při uvedení argumentu PZ jsou vypočítány nuly i póly najednou.

### 3.1.4 Vliv teploty

U všech součástek zadávaných při popisu obvodu se předpokládá, že jejich parametry byly změřeny při nominální teplotě, která je rovna 300 K (27° C). Všechny analýzy jsou standardně prováděny při této nominální teplotě.

### Teplotní analýza TEMP

Obecný tvar:

```
TEMP TEMP1 <TEMP2 <TEMP3 <...>>>
```

Příklady:

```
TEMP 27
TEMP 0 20 40
```


Specifikuje seznam teplot ve stupních Celsia. Následující analýzy budou prováděny opakovaně pro všechny uvedené teploty. Při vykreslení grafu budou zobrazeny přes sebe výsledky pro všechny uvedené teploty.

## 4 Příklady obvodů v programu WinSpice

### Příklad 1

Obvod uvedený na obrázku má parametry  $R_1 = 1 \text{ k}\Omega$ ,  $R_2 = 2 \text{ k}\Omega$ ,  $R_3 = 3 \text{ k}\Omega$ ,  $R_4 = 4 \text{ k}\Omega$ ,  $U = 15 \text{ V}$ ,  $I = 5 \text{ mA}$  a je ve stacionárním ustáleném stavu SUS. Vypočítejte SUS pomocí OP analýzy a zobrazte napětí na rezistorech  $R_2$  a  $R_4$  a na proudovém zdroji  $I_1$  dále zobrazte proud v napěťovém zdroji  $U_1$ . Zobrazte také výkon dodávaný oběma zdroji.<sup>3</sup>

<sup>3</sup>WinSpice zobrazuje u napěťového zdroje výkon dodávaný do obvodu s kladným znaménkem a u proudového zdroje se záporným znaménkem.




---

```


Obvod na SUS
I1 0 1 5m
V1 3 0 15
R1 1 2 1K
R2 2 0 2K
R3 2 3 3K
R4 1 3 4K
.CONTROL
DESTROY ALL
OP
PRINT V(2) V(1,3) V(1) I(V1)
PRINT -@I1[P] @V1[P]
.ENDC
.END

```

---

## Příklad 2

Obvod uvedený na obrázku má parametry  $R_1 = 1\text{ k}\Omega$ ,  $R_2 = 1\text{ k}\Omega$ ,  $C = 1\text{ }\mu\text{F}$  a  $L = 1\text{ H}$ . Nejprve předpokládejte, že je obvod připojen ke zdroji stejnosměrného napětí a je v SUS. Pomocí TF analýzy vypočítejte přenos, vstupní a výstupní impedanci. Výsledky výpočtu zobrazte. Dále předpokládejte, že je obvod připojen ke zdroji harmonického napětí a je v harmonickém ustáleném stavu HUS. Pomocí AC analýzy vypočítejte kmitočtovou charakteristiku v intervalu 10 Hz až 10 kHz a vykreslete závislost modulu a fáze na kmitočtu.



---

```

Obvod na HUS
V1 1 0 1 AC 1
L1 1 2 1
R1 2 3 1K
R2 3 0 1K
C1 3 0 1U
.CONTROL
DESTROY ALL
SET MAXPLOTS=2
TF v(3) V1
PRINT ALL
AC DEC 10 10 10k
PLOT DB(V(3)) XLIMIT 10 10K
PLOT PHASE(V(3)) XLIMIT 10 10K
.ENDC
.END

```

---

### Příklad 3

Obvod z příkladu 2 je připojen na zdroj s pulsním průběhem napětí podle funkce `PULSE(0 1)`. Pomocí přechodové `TRAN` analýzy vypočítejte časovou odezvu obvodu na intervalu od 0 do 20 ms s krokem  $100 \mu\text{s}$ . Dále použitím příkazu `ALTER` změňte hodnotu rezistoru  $R_1$  na  $100 \Omega$  a opakujte přechodovou analýzu. Vykreslete časovou závislost vstupního a výstupního napětí pro obě hodnoty rezistoru do jednoho grafu.

---

```

Prechodova analyza - ss zdroj
V1 1 0 0 PULSE(0 1)
L1 1 2 1
R1 2 3 1K
R2 3 0 1K
C1 3 0 1U
.CONTROL
DESTROY ALL
SET MAXPLOTS=2
TRAN 100U 20M
ALTER R1=100
TRAN 100U 20M
PLOT V(1) TRAN1.V(3) TRAN2.V(3)
.ENDC
.END

```

---

### Příklad 4

Uvažujte stejný obvod jako v předchozím příkladu s tím rozdílem, že se na vstup připojen zdroj harmonického napětí s kmitočtem  $f = 400 \text{ Hz}$  a amplitudou  $U = 1 \text{ V}$ . Zdroj začne působit až v čase  $t = 0$  (použijte funkci `SIN(0 1 400)`). Pomocí přechodové `TRAN` analýzy vypočítejte časovou odezvu obvodu na intervalu od 0 do 20 ms s krokem  $100 \mu\text{s}$ . Vykreslete časovou závislost vstupního a výstupního napětí. Změňte kmitočet zdroje na  $f = 200 \text{ Hz}$  a znovu proveďte přechodovou analýzu pro stejný časový interval. Vykreslete časovou závislost výstupního napětí pro oba kmitočty.



---

```


Prechodova analyza - st zdroj
V1 1 0 0 SIN(0 1 400)
L1 1 2 1
R1 2 3 100
R2 3 0 1K
C1 3 0 1U
.CONTROL
DESTROY ALL
SET MAXPLOTS=2
TRAN 100U 20M
PLOT V(1) V(3)
ALTER @V1[SIN]=[0 1 200]
TRAN 100U 20M
PLOT TRAN1.V(3) TRAN2.V(3)
.ENDC
.END

```

---

### Příklad 5

Obvod uvedený na obrázku má parametry  $U_1 = 2\text{V}$ ,  $R_1 = 1\text{k}\Omega$  a je ve stacionárním ustáleném stavu. Nalezněte pracovní bod pomocí OP analýzy a výsledky výpočtu zobrazte pomocí příkazu SHOW. Použijte model diody s výchozími parametry. Parametry modelu diody zobrazte pomocí příkazu SHOWMOD.




---

```

Obvod s diodou v SUS
V1 1 0 2
R1 1 2 1K
D1 2 0 DIODA
.MODEL DIODA D
.CONTROL
DESTROY ALL
OP
SHOW d1
SHOWMOD
*ve verzi 1.6.0 nutno pouzít SHOW D1
.ENDC
.END

```

---

### Příklad 6

Pro obvod z příkladu 5 proveďte stejnosměrnou analýzu pokud se napětí zdroje  $U_1$  mění v rozsahu -5 až 5 V. Vykreslete VA charakteristiku diody jako závislost proudu zdroje  $U_1$  na napětí na diodě  $D_1$ . Změňte průrazné napětí v závěrném směru v modelu diody na hodnotu  $BV = 3\text{V}$  a celý postup opakujte.

---

VA charakteristika diody

```
V1 1 0 0
R1 1 2 1k
D1 2 0 DIODA
.MODEL DIODA D
.CONTROL
DESTROY ALL
SET MAXPLOTS=2
DC V1 -5 5 10m
PLOT -I(V1) VS V(2)
ALTER @DIODA[BV]=3
DC V1 -5 5 10m
PLOT -I(V1) VS V(2)
.ENDC
.END
```

---

### Příklad 7

Zadání příkladu 6 řešte pro tři různé teploty 0°C, 27°C a 100°C.

---

VA charakteristika diody - teplotní závislost

```
V1 1 0 0
R1 1 2 1k
D1 2 0 DIODA
.MODEL DIODA D
.CONTROL
TEMP 0 27 100
DC V1 -5 5 10m
PLOT -I(V1) VS V(2)
.ENDC
.END
```


---

### Příklad 8

Obvod uvedený na obrázku má parametry  $R_1 = 1 \text{ k}\Omega$ ,  $R_2 = 1 \text{ M}\Omega$ ,  $C_1 = C_2 = 100 \mu\text{F}$ . Dioda  $D_1$  má výchozí parametry. Na vstup je připojen zdroj harmonického napětí  $u_1$  s amplitudou 1 mV a kmitočtem 100 Hz. Pomocí přechodové analýzy vypočítejte časový průběh napětí  $u_4$  pro tři různé hodnoty proudu  $I_1 = 1 \text{ mA}$ ,  $0.1 \text{ mA}$  a  $10 \mu\text{A}$ . Porovnejte amplitudu výstupního napětí s přenosem vypočítaným podle vztahu

$$P = \frac{r_d}{r_d + R_1},$$

kde  $r_d$  je difuzní odpor diody v pracovním bodě.




---

```

DIODA TRAN ANALYZA
V1 1 0 0 SIN(0 1m 100) AC 1
R1 1 2 1k
C1 2 3 100U
I1 0 3 1m
D1 3 0 DIODA
C2 3 4 100U
R2 4 0 1MEG
.MODEL DIODA D
.CONTROL
DESTROY ALL
SET MAXPLOTS=1
OP
LET prenos=1/(1+1e3*(@D1[gd]))
PRINT prenos
SET MAXPLOTS=1
TRAN 100U 50m
PLOT V(4)
.ENDC

.END

```

---

### Příklad 9

Obvod z předchozího příkladu s  $I_1 = 10 \mu\text{A}$  je buzen zdrojem harmonického napětí  $u_1$  o kmitočtu 100 Hz. Uvažujte amplitudu postupně 1 mV, 10 mV a 100 mV. Pomocí přechodové analýzy vypočítejte časový průběh napětí  $u_4$ . Na časovém průběhu výstupního napětí proveďte Fourierovu analýzu. Porovnejte hodnotu celkového harmonického zkreslení THD pro jednotlivé amplitudy.

---

```

DIODA FOURIEROVA ANALYZA
V1 1 0 0 SIN(0 1m 100) AC 1
R1 1 2 1k
C1 2 3 100U
I1 0 3 100U
D1 3 0 DIODA
C2 3 4 100U
R2 4 0 1MEG
.MODEL DIODA D
.CONTROL
DESTROY ALL
SET MAXPLOTS=1
TRAN 100U 50m
PLOT V(4)
FOURIER 100 V(4)
.ENDC

.END

```

---

### Příklad 10

Zobrazte VA charakteristiky bipolárního tranzistoru NPN v zapojení se společným emitorem. Použijte model tranzistoru s výchozími parametry a dále s hodnotou VAF=50 (Earlyho napětí v aktivním režimu).

---

```

VA charakteristika bipolarniho tranzistoru SE
V1 1 0 0
I1 0 2 0
Q1 1 2 0 NPN1
.MODEL NPN1 NPN
.CONTROL
DESTROY ALL
SET MAXPLOTS=1
DC V1 0 5 10M I1 0 1M 200U
ALTER @NPN1[VAF]=50
DC V1 0 5 10M I1 0 1M 200U
PLOT -DC1.V1#BRANCH -DC2.V1#BRANCH
.ENDC


.END

```

---

### Příklad 11

Zapojení na obrázku má parametry  $U_{cc} = 10\text{ V}$ ,  $R_c = 5\text{ k}\Omega$ ,  $R_b = 910\text{ k}\Omega$ ,  $R_z = 1\text{ M}\Omega$ ,  $C_{in} = 10\text{ }\mu\text{F}$ ,  $C_{out} = 10\text{ }\mu\text{F}$ .  $u_1(t)$  je zdroj harmonického napětí s amplitudou 1 mV a frekvencí 1 kHz. Tranzistor  $T_1$  má výchozí parametry. Proveďte výpočet pracovního bodu pomocí OP analýzy a výsledky zobrazte pomocí příkazu SHOW. Pomocí AC analýzy vypočítejte kmitočtovou charakteristiku v intervalu 1 Hz až 10 kHz a vykreslete závislost modulu napětí na rezistoru  $R_z$  na kmitočtu. Pro AC analýzu uvažujte vstupní napětí 1 V. Pomocí přechodové analýzy vypočítejte časový průběh napětí na rezistoru  $R_z$ . Na časovém průběhu napětí proveďte Fourierovu analýzu.




---

Zesilovac s NPN tranzistorem v zapojeni SE

```

Vcc 1 0 10
Rc 1 2 5k
Rb 1 3 910k
Q1 2 3 0 NPN1
C1 4 3 10u
C2 2 5 10u
V1 4 0 0 SIN(0 1m 1k) AC 1
Rz 5 0 1Meg
.MODEL NPN1 NPN
.CONTROL
DESTROY ALL
SET MAXPLOTS=2
OP
SHOW q1
AC DEC 10 1 10k
plot vdb(5)
TRAN 10U 5M
plot v(5)
fourier 1k v(5)
.ENDC
.END

```

---

## Reference

- [1] Smith, M.: *WinSpice3 User's Manual*. Součást instalace programu WinSpice3
- [2] Smith, M.: <http://www.winspice.com/>